
Exercises werkcollege 7 
 

Exercise 1 
A self-service gas station has a number of pumps for delivering gas to customers for their 
vehicles. Customers are expected to pre-pay a cashier for their gas. The cashier activates 
a pump to deliver gas. 
 

a) Provide a model for a simple system with 2 customers and a gas station with 1 
pump and a cashier. Include into the model a range for the different amounts of 
payment and corresponding amounts of gas. A customer is not satisfied if an 
incorrect amount of gas is delivered; this should be reflected by an ERROR state 
of the model. 

 
const N = 2 
range C = 1..N 
range G = 1..3 

 
CUSTOMER = (prepay[a:G] ->  

             pump[b:G] -> 
                if a==b then CUSTOMER 
                             else ERROR). 
 

||CUSTOMERS = c[C]:CUSTOMER. 
 

CASHIER = (c[i:C].prepay[a:G] -> c[i].deliver[a] -> CASHIER). 
 

PUMP = (c[k:C].deliver[x:G] -> c[k].pump[x] -> PUMP). 
 

||STATION = (CUSTOMERS || CASHIER || PUMP) 
               \{c[C].deliver}. 
 

b) Check the safety and progress properties for this system.  
 

No deadlocks/errors 
No progress violations detected. 

 
c) Extend the model to cater for 2 customers and 2 pumps. Specify and check a 

safety property FIFO that ensures that customers are served in the order in which 
they pay. 

 
Add and modify as follows: 
 
const M = 2 
range P = 1..M 
 
CUSTOMER = (prepay[a:G] ->  
               p[P].pump[b:G] -> 
                  if a==b then CUSTOMER 
                          else ERROR). 



 
||CUSTOMERS = c[C]:CUSTOMER. 
 
CASHIER = (c[i:C].prepay[a:G] -> 
              p[P].c[i].deliver[a] ->  
                 CASHIER). 
 
PUMP = (c[k:C].deliver[x:G] -> c[k].pump[x] -> PUMP). 
||PUMPS = p[P]:PUMP. 
 
        
||STATION = (CUSTOMERS || CASHIER || PUMPS) 
          /{forall[m:C][n:P] {c[m].p[n].pump/p[n].c[m].pump}} 

\{p[P].c[C].deliver}. 
 

 

       d) 
property FIFO = (c[i:C].prepay[G] -> FIFO[i]), 
FIFO[i:C] = (c[i].p[P].pump[G] -> FIFO | c[j:C].prepay[G] -> 
FIFO[i][j]), 
FIFO[i:C][j:C] = (c[i].p[P].pump[G] -> FIFO[j]). 
       

 
||SAFE_STATION = (STATION || FIFO). 

 
The FIFO property does not hold (which is a fortunate thing. Imagine a customer 
who pays, and then first takes a nap before pumping…. ) The FIFO property does 
hold w.r.t. the deliver actions. 
 


